Pensieve header: The full sl_2 invariant using the Drinfel'd double. Continues 2018-05/ybax.nb, Talks/StonyBrook-1805/ybax.nb, Projects/SL2Portfolio/Logoi.nb.

Profiling

In[∘]:=
```
Once[
  SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\SL2Invariant"];
  << KnotTheory`;
  << "../Profile/Profile.m";
 ];
BeginProfile[];
Once@PopupWindow[Button["Show Profile Monitor"],
  Dynamic[PrintProfile[], UpdateInterval → 3, TrackedSymbols → {}]]
```

```
Loading KnotTheory` version of January 20, 2015, 10:42:19.1122.
Read more at http://katlas.org/wiki/KnotTheory.
This is Profile.m of http://www.drorbn.net/AcademicPensieve/Projects/Profile/.
This version: June 2018. Original version: July 1994.
```

Out[∘]= Show Profile Monitor

External Utilities

In[∘]:=
```
HL[ℰ_] := Style[ℰ, Background → Yellow];
```

Program

Program

Internal Utilities

Program

Canonical Form:

Program

In[∘]:=
```
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := PP_CF@ExpandDenominator@ExpandNumerator@PP_Together@Together[PP_Exp[
        Expand[ℰ] //. ⅇ^x_ ⅇ^y_ :> ⅇ^(x+y) /. ⅇ^x_ :> ⅇ^CF[x]]];
```

Program

The Kronecker δ:

Program

In[∘]:=
```
Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

Program

Equality, multiplication, and degree-adjustment of perturbed Gaussians; $\mathbb{E}[L, Q, P]$ stands for $e^{L+Q} P$:

Program

In[∘]:=
```
𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
   CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
𝔼[L_, Q_, P_]$k_ := 𝔼[L, Q, Series[Normal@P, {ϵ, 0, $k}]];
```

Program

Zip and Bind

Program

Variables and their duals:

Program

In[∘]:=
```
{t*, b*, y*, a*, x*, z*} = {τ, β, η, α, ξ, ζ};
{τ*, β*, η*, α*, ξ*, ζ*} = {t, b, y, a, x, z}; (u_i_)* := (u*)i;
```

Program

Finite Zips:

Program

In[∘]:=
```
collect[sd_SeriesData, ζ_] := MapAt[collect[#, ζ] &, sd, 3];
collect[ℰ_, ζ_] := PPCollect@Collect[ℰ, ζ];
Zip{}[P_] := P; Zip{ζ_,ζs___}[P_] := PPZip[
   (collect[P // Zip{ζs}, ζ] /. f_. ζ^d_. :→ ∂{ζ*,d} f) /. ζ* → 0]
```

Program

QZip implements the "Q-level zips" on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard the L variables as scalars.

Program

In[∘]:=
```
QZipζs_List@𝔼[L_, Q_, P_] := PPQZip@Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
     zs = Table[ζ*, {ζ, ζs}];
     c = Q /. Alternatives @@ (ζs ⋃ zs) → 0;
     ys = Table[∂ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
     ηs = Table[∂z (Q /. Alternatives @@ ζs → 0), {z, zs}];
     qt = Inverse@Table[Kδz,ζ* - ∂z,ζ Q, {ζ, ζs}, {z, zs}];
     zrule = Thread[zs → qt.(zs + ys)];
     Q2 = (Q1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
     CF /@ 𝔼[L, Q2, Det[qt] e^-Q2 Zipζs[e^Q1 (P /. zrule)]] ];
```

Program

Upper to lower and lower to Upper:

Program

In[∘]:=
```
U2l = {Bi_^p_. → e^(-p ħ γ bi), B^p_. → e^(-p ħ γ b), Ti_^p_. → e^(p ħ ti), T^p_. → e^(p ħ t), 𝒜i_^p_. → e^(p γ αi), 𝒜^p_. → e^(p γ α)};
l2U = {e^(c_. bi_+d_.) :→ Bi^(-c/(ħ γ)) e^d, e^(c_. b+d_.) :→ B^(-c/(ħ γ)) e^d,
    e^(c_. ti_+d_.) :→ Ti^(c/ħ) e^d, e^(c_. t+d_.) :→ T^(c/ħ) e^d,
    e^(c_. αi_+d_.) :→ 𝒜i^(c/γ) e^d, e^(c_. α+d_.) :→ 𝒜^(c/γ) e^d,
    e^ℰ_ :→ e^(Expand@ℰ)};
```

Program

LZip implements the "L-level zips" on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard all of Pe^Q as a single"P". Here the z's are b and α and the ζ's are β and a.

Program

```
In[∘]:= LZip_{ξs_List}@𝔼[L_, Q_, P_] := PP_{LZip}@Module[{ξ, z, zs, c, ys, ηs, lt, zrule, L1, L2, Q1, Q2},
    zs = Table[ξ*, {ξ, ξs}];
    c = L /. Alternatives @@ (ξs ⋃ zs) → 0;
    ys = Table[∂_ξ (L /. Alternatives @@ zs → 0), {ξ, ξs}];
    ηs = Table[∂_z (L /. Alternatives @@ ξs → 0), {z, zs}];
    lt = Inverse@Table[Kδ_{z,ξ*} - ∂_{z,ξ} L, {ξ, ξs}, {z, zs}];
    zrule = Thread[zs → lt.(zs + ys)];
    L2 = (L1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
    Q2 = (Q1 = Q /. U2l /. zrule) /. Alternatives @@ zs → 0;
    CF /@ 𝔼[L2, Q2, Det[lt] e^{-L2-Q2} Zip_{ξs}[e^{L1+Q1} (P /. U2l /. zrule)]] //. l2U];
```

Program

```
In[∘]:= Bind_{}[L_, R_] := L R;
Bind_{is__}[L_𝔼, R_𝔼] := PP_{Bind}@Module[{n},
    Times[
      L /. Table[(v : b | B | t | T | a | x | y)_i → v_{n@i}, {i, {is}}],
      R /. Table[(v : β | τ | α | 𝒜 | ξ | η)_i → v_{n@i}, {i, {is}}]
    ] // LZip_{Flatten@Table[{β_{n@i},τ_{n@i},a_{n@i}},{i,{is}}]} // QZip_{Flatten@Table[{ξ_{n@i},y_{n@i}},{i,{is}}]}];
B_{l_List}[L_, R_] := Bind_l[L, R]; B_{is___}[L_, R_] := Bind_{{is}}[L, R];
```

Program

“Define” code

Program

Define[lhs = rhs,   ...] defines the lhs to be rhs, except that rhs is computed only once for each value of $k. Fancy Mathematica not for the faint of heart. Most readers should ignore.

Program

```
In[∘]:= SetAttributes[Define, HoldAll];
Define[def_, defs__] := (Define[def]; Define[defs];);
Define[op_{is__} = ℰ_] := Module[{SD, ii, jj, kk, isp, nis, nisp, sis}, Block[{i, j, k},
    ReleaseHold[Hold[
        SD[op_{nisp,$k_Integer}, PP_{Boot@$k}@Block[{i, j, k}, op_{isp,$k} = ℰ; op_{nis,$k}]];
        SD[op_{isp}, op_{{is},$k}]; SD[op_{sis__}, op_{{sis}}];
      ] /. {SD → SetDelayed,
        isp → {is} /. {i → i_, j → j_, k → k_},
        nis → {is} /. {i → ii, j → jj, k → kk},
        nisp → {is} /. {i → ii_, j → jj_, k → kk_}
      }] ]]
```

Program

Booting Up

Program

```
In[∘]:= $k = 2; ℏ = γ = 1;
```

Program

In[∘]:=

$$\mathtt{Define}\Big[\mathtt{am}_{i,j\to k} = \mathbb{E}\big[(\alpha_i+\alpha_j)\,a_k,\ \big(e^{-\gamma\,\alpha_j}\,\xi_i+\xi_j\big)\,x_k,\ 1\big]_{\$k},\\ \mathtt{bm}_{i,j\to k} = \mathbb{E}\Big[(\beta_i+\beta_j)\,b_k,\ (\eta_i+\eta_j)\,y_k,\ e^{\left(e^{-\epsilon\,\beta_i}-1\right)\,\eta_j\,y_k}\Big]_{\$k}\Big]$$

Program

In[∘]:=

$$\mathtt{Define}\Big[R_{i,j} = \mathbb{E}\Big[\hbar\,a_j\,b_i,\ \hbar\,x_j\,y_i,\ e\,\verb|^|\Big(\sum_{k=2}^{\$k+1}\frac{\left(1-e^{\gamma\,\epsilon\,\hbar}\right)^k\,(\hbar\,y_i\,x_j)^k}{k\,\left(1-e^{k\,\gamma\,\epsilon\,\hbar}\right)}\Big)\Big]_{\$k},\\ P_{i,j} = \mathbb{E}\big[\beta_i\,\alpha_j/\hbar,\ \eta_i\,\xi_j/\hbar,\\ 1+\mathtt{If}\big[\$k == 0,\ 0,\ \mathtt{Normal}@P_{\{i,j\},\$k-1}[[3]] - \big(R_{1,2}\sim B_{1,2}\sim\big((P_{\{1,j\},0})_{\$k}\ (P_{\{i,2\},\$k-1})_{\$k}\big)\big)[[3]]\big]\big]\Big]$$

Program

In[∘]:=

$$\mathtt{Define}\Big[aS_i = \mathbb{E}\Big[-\alpha_i\,a_j,\ -\xi_i\,x_i,\\ \mathtt{Sum}\Big[\mathtt{Expand}\Big[\frac{e^{\xi_i\,x_i}\,(-\hbar\,\gamma\,\epsilon)^k}{2^k\,k\,!}\,\mathtt{Nest}\Big[\mathtt{Expand}\big[x_i^2\,\partial_{\{x_i,2\}}\,\#\big]\ \&,\ e^{-\xi_i\,e^{\hbar\,\epsilon\,a_i}\,x_i},\ k\Big]\Big],\ \{k,0,\$k\}\Big]\Big]_{\$k}\sim\\ B_{i,j}\sim \mathtt{am}_{i,j\to i},\\ \overline{aS}_i = \mathbb{E}\Big[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\mathtt{If}\Big[\$k == 0,\ 0,\\ \mathtt{Normal}@\overline{aS}_{\{i\},\$k-1}[[3]] - \Big(\big(\overline{aS}_{\{i\},0}\big)_{\$k}\sim B_i\sim aS_i\sim B_i\sim\big(\overline{aS}_{\{i\},\$k-1}\big)_{\$k}\Big)[[3]]\Big]\Big]\Big]$$

Program

In[∘]:=

$$\mathtt{Define}\Big[bS_i = R_{i,1}\sim B_1\sim aS_1\sim B_1\sim P_{i,1},\\ \overline{bS}_i = R_{i,1}\sim B_1\sim\overline{aS}_1\sim B_1\sim P_{i,1},\\ a\Delta_{i\to j,k} = (R_{1,j}\,R_{2,k})\sim B_{1,2}\sim \mathtt{bm}_{1,2\to 3}\sim B_3\sim P_{3,i},\\ b\Delta_{i\to j,k} = (R_{j,1}\,R_{k,2})\sim B_{1,2}\sim \mathtt{am}_{1,2\to 3}\sim B_3\sim P_{i,3}\Big]$$

Program

In[∘]:=

$$\mathtt{Define}\Big[\mathtt{dm}_{i,j\to k} =\\ \Big(\mathbb{E}[\beta_i\,b_i+\alpha_j\,a_j,\ \eta_i\,y_i+\xi_j\,x_j,\ 1]\ \big(a\Delta_{i\to 1,2}\sim B_2\sim a\Delta_{2\to 2,3}\sim B_3\sim\overline{aS}_3\big)\ \big(b\Delta_{j\to -1,-2}\sim B_{-2}\sim b\Delta_{-2\to -2,-3}\big)\Big)\sim\\ B_{-3,-2,-1,1,2,3,i,j}\sim\big(P_{-1,3}\,P_{-3,1}\,\mathtt{am}_{2,j\to k}\,\mathtt{bm}_{i,-2\to k}\big),\\ dS_i = \mathbb{E}[\beta_i\,b_1+\alpha_i\,a_2,\ \eta_i\,y_1+\xi_i\,x_2,\ 1]\sim B_{1,2}\sim\big(\overline{bS}_1\,aS_2\big)\sim B_{1,2}\sim \mathtt{dm}_{2,1\to i},\\ d\Delta_{i\to j,k} = \big(b\Delta_{i\to 3,1}\,a\Delta_{i\to 2,4}\big)\sim B_{1,2,3,4}\sim\big(\mathtt{dm}_{3,4\to k}\,\mathtt{dm}_{1,2\to j}\big)\Big]$$

Program

In[∘]:=

$$\mathtt{Define}\Big[\overline{R}_{i,j} = \mathtt{Expand}\ /@\ R_{i,j}\sim B_j\sim dS_j,\\ C_i = \mathbb{E}\big[0,\ 0,\ B_i^{1/2}\,e^{-\hbar\,\epsilon\,a_i/2}\big]_{\$k},\\ \overline{C}_i = \mathbb{E}\big[0,\ 0,\ B_i^{-1/2}\,e^{\hbar\,\epsilon\,a_i/2}\big]_{\$k},\\ \mathtt{Kink}_i = \big(R_{1,3}\,\overline{C}_2\big)\sim B_{1,2}\sim \mathtt{dm}_{1,2\to 1}\sim B_{1,3}\sim \mathtt{dm}_{1,3\to i},\\ \overline{\mathtt{Kink}}_i = \big(\overline{R}_{1,3}\,C_2\big)\sim B_{1,2}\sim \mathtt{dm}_{1,2\to 1}\sim B_{1,3}\sim \mathtt{dm}_{1,3\to i}\Big]$$

Program

Note. $t == \epsilon \mathrm{a} - \gamma \mathrm{b}$ and $b == -t/\gamma + \epsilon \mathrm{a}/\gamma$.

Program

In[∘]:=

$$\mathtt{Define}\Big[\mathtt{b2t}_i = \mathbb{E}\big[\alpha_i\,a_i-\beta_i\,t_i\big/\gamma,\ \xi_i\,x_i+\eta_i\,y_i,\ e^{\epsilon\,\beta_i\,a_i/\gamma}\big]_{\$k},\\ \mathtt{t2b}_i = \mathbb{E}\big[\alpha_i\,a_i-\tau_i\,\gamma\,b_i,\ \xi_i\,x_i+\eta_i\,y_i,\ e^{\epsilon\,\tau_i\,a_i}\big]_{\$k}\Big]$$

Testing

In[∘]:=
```
Block[{$k = 1}, {
   am → am_{i,j→k}, bm → bm_{i,j→k}, dm → dm_{i,j→k}, R → R_{i,j}, R̄ → R̄_{i,j}, P → P_{i,j},
   aS → aS_i, aS̄ → aS̄_i, bS → bS_i, bS̄ → bS̄_i, dS → dS_i, aΔ → aΔ_{i→j,k}, bΔ → bΔ_{i→j,k},
   dΔ → dΔ_{i→j,k}, C → C_i, C̄ → C̄_i, Kink → Kink_i, Kink̄ → Kink̄_i, b2t → b2t_i, t2b → t2b_i
  }] //
 Column
```

Out[∘]=

$$\mathrm{am} \to \mathbb{E}\left[a_k\,(\alpha_i+\alpha_j),\ x_k\left(e^{-\alpha_j}\xi_i+\xi_j\right),\ 1\right]$$

$$\mathrm{bm} \to \mathbb{E}\left[b_k\,(\beta_i+\beta_j),\ y_k\,(\eta_i+\eta_j),\ 1-y_k\,\beta_i\,\eta_j\,\epsilon+O[\epsilon]^2\right]$$

$$\mathrm{dm} \to \mathbb{E}\Big[a_k\,\alpha_i+a_k\,\alpha_j+b_k\,\beta_i+b_k\,\beta_j,\ \frac{1}{\mathcal{A}_i\,\mathcal{A}_j}\left(y_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_i+y_k\,\mathcal{A}_j\,\eta_j+x_k\,\mathcal{A}_i\,\xi_i+\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i-B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i+x_k\,\mathcal{A}_i\,\mathcal{A}_j\,\xi_j\right),\ 1+\frac{1}{4\,\mathcal{A}_i\,\mathcal{A}_j}\Big(-4\,y_k\,\mathcal{A}_j\,\beta_i\,\eta_j-4\,x_k\,\mathcal{A}_i\,\beta_j\,\xi_i+4\,x_k\,y_k\,\eta_j\,\xi_i+4\,a_k\,B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j\,\xi_i+2\,y_k\,\mathcal{A}_j\,\eta_j^2\,\xi_i-6\,B_k\,y_k\,\mathcal{A}_j\,\eta_j^2\,\xi_i+2\,x_k\,\mathcal{A}_i\,\eta_j\,\xi_i^2-6\,B_k\,x_k\,\mathcal{A}_i\,\eta_j\,\xi_i^2+\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2-4\,B_k\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2+3\,B_k^2\,\mathcal{A}_i\,\mathcal{A}_j\,\eta_j^2\,\xi_i^2\Big)\,\epsilon+O[\epsilon]^2\Big]$$

$$\mathrm{R} \to \mathbb{E}\left[a_j\,b_i,\ x_j\,y_i,\ 1-\frac{1}{4}\left(x_j^2\,y_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{\mathrm{R}} \to \mathbb{E}\left[-a_j\,b_i,\ -\frac{x_j\,y_i}{B_i},\ 1+\frac{\left(-4\,a_j\,B_i\,x_j\,y_i-3\,x_j^2\,y_i^2\right)\epsilon}{4\,B_i^2}+O[\epsilon]^2\right]$$

$$\mathrm{P} \to \mathbb{E}\left[\alpha_j\,\beta_i,\ \eta_i\,\xi_j,\ 1+\frac{1}{4}\,\eta_i^2\,\xi_j^2\,\epsilon+O[\epsilon]^2\right]$$

$$\mathrm{aS} \to \mathbb{E}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\frac{1}{2}\left(-2\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{\mathrm{aS}} \to \mathbb{E}\left[-a_i\,\alpha_i,\ -x_i\,\mathcal{A}_i\,\xi_i,\ 1+\frac{1}{2}\left(2\,x_i\,\mathcal{A}_i\,\xi_i-2\,a_i\,x_i\,\mathcal{A}_i\,\xi_i-x_i^2\,\mathcal{A}_i^2\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\mathrm{bS} \to \mathbb{E}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1+\frac{\left(-2\,B_i\,y_i\,\beta_i\,\eta_i-y_i^2\,\eta_i^2\right)\epsilon}{2\,B_i^2}+O[\epsilon]^2\right]$$

$$\overline{\mathrm{bS}} \to \mathbb{E}\left[-b_i\,\beta_i,\ -\frac{y_i\,\eta_i}{B_i},\ 1+\frac{\left(2\,B_i\,y_i\,\eta_i-2\,B_i\,y_i\,\beta_i\,\eta_i-y_i^2\,\eta_i^2\right)\epsilon}{2\,B_i^2}+O[\epsilon]^2\right]$$

$$\mathrm{dS} \to \mathbb{E}\Big[-a_i\,\alpha_i-b_i\,\beta_i,\ \frac{-y_i\,\mathcal{A}_i\,\eta_i-B_i\,x_i\,\mathcal{A}_i\,\xi_i+\mathcal{A}_i\,\eta_i\,\xi_i-B_i\,\mathcal{A}_i\,\eta_i\,\xi_i}{B_i},\ 1+\frac{1}{4\,B_i^2}\Big(4\,B_i\,y_i\,\mathcal{A}_i\,\eta_i-4\,B_i\,y_i\,\mathcal{A}_i\,\beta_i\,\eta_i-2\,y_i^2\,\mathcal{A}_i^2\,\eta_i^2-4\,a_i\,B_i^2\,x_i\,\mathcal{A}_i\,\xi_i-4\,B_i^2\,x_i\,\mathcal{A}_i\,\beta_i\,\xi_i-4\,B_i\,\mathcal{A}_i\,\eta_i\,\xi_i+4\,a_i\,B_i\,\mathcal{A}_i\,\eta_i\,\xi_i+4\,B_i^2\,\mathcal{A}_i\,\eta_i\,\xi_i-4\,B_i\,x_i\,y_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i+4\,B_i\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i-4\,B_i^2\,\mathcal{A}_i\,\beta_i\,\eta_i\,\xi_i+6\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i-2\,B_i\,y_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i-2\,B_i^2\,x_i^2\,\mathcal{A}_i^2\,\xi_i^2+6\,B_i\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2-2\,B_i^2\,x_i\,\mathcal{A}_i^2\,\eta_i\,\xi_i^2-3\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2+4\,B_i\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2-B_i^2\,\mathcal{A}_i^2\,\eta_i^2\,\xi_i^2\Big)\,\epsilon+O[\epsilon]^2\Big]$$

$$\mathrm{a}\Delta \to \mathbb{E}\left[a_j\,\alpha_i+a_k\,\alpha_i,\ x_j\,\xi_i+x_k\,\xi_i,\ 1+\frac{1}{2}\left(-2\,a_j\,x_k\,\xi_i+x_j\,x_k\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\mathrm{b}\Delta \to \mathbb{E}\left[b_j\,\beta_i+b_k\,\beta_i,\ B_k\,y_j\,\eta_i+y_k\,\eta_i,\ 1+\frac{1}{2}\,B_k\,y_j\,y_k\,\eta_i^2\,\epsilon+O[\epsilon]^2\right]$$

$$\mathrm{d}\Delta \to \mathbb{E}\left[a_j\,\alpha_i+a_k\,\alpha_i+b_j\,\beta_i+b_k\,\beta_i,\ y_j\,\eta_i+B_j\,y_k\,\eta_i+x_j\,\xi_i+x_k\,\xi_i,\ 1+\frac{1}{2}\left(B_j\,y_j\,y_k\,\eta_i^2-2\,a_j\,x_k\,\xi_i+x_j\,x_k\,\xi_i^2\right)\epsilon+O[\epsilon]^2\right]$$

$$\mathrm{C} \to \mathbb{E}\left[0,\ 0,\ \sqrt{B_i}-\frac{1}{2}\left(a_i\,\sqrt{B_i}\right)\epsilon+O[\epsilon]^2\right]$$

$$\overline{\mathrm{C}} \to \mathbb{E}\left[0,\ 0,\ \frac{1}{\sqrt{B_i}}+\frac{a_i\,\epsilon}{2\,\sqrt{B_i}}+O[\epsilon]^2\right]$$

$$\mathrm{Kink} \to \mathbb{E}\left[a_i\,b_i,\ x_i\,y_i,\ \frac{1}{\sqrt{B_i}}+\frac{\left(2\,a_i-x_i^2\,y_i^2\right)\epsilon}{4\,\sqrt{B_i}}+O[\epsilon]^2\right]$$

$$\overline{\mathrm{Kink}} \to \mathbb{E}\left[-a_i\,b_i,\ -\frac{x_i\,y_i}{B_i},\ \sqrt{B_i}+\frac{\left(-2\,a_i\,B_i^2-4\,a_i\,B_i\,x_i\,y_i-3\,x_i^2\,y_i^2\right)\epsilon}{4\,B_i^{3/2}}+O[\epsilon]^2\right]$$

$$\mathrm{b2t} \to \mathbb{E}\left[a_i\,\alpha_i-t_i\,\beta_i,\ y_i\,\eta_i+x_i\,\xi_i,\ 1+a_i\,\beta_i\,\epsilon+O[\epsilon]^2\right]$$

$$\mathrm{t2b} \to \mathbb{E}\left[a_i\,\alpha_i-b_i\,\tau_i,\ y_i\,\eta_i+x_i\,\xi_i,\ 1+a_i\,\tau_i\,\epsilon+O[\epsilon]^2\right]$$

Check that on the generators this agrees with our conventions in the handout:

```
In[∘]:= Timing@{{"[a,x]" → ((𝔼[0, 0, a₂ x₁] ~B₁,₂~ am₁,₂→₁)⟦3⟧ - (𝔼[0, 0, a₁ x₂] ~B₁,₂~ am₁,₂→₁)⟦3⟧),
      "[b,y]" → ((𝔼[0, 0, y₂ b₁] ~B₁,₂~ bm₁,₂→₁)⟦3⟧ - (𝔼[0, 0, y₁ b₂] ~B₁,₂~ bm₁,₂→₁)⟦3⟧)} /.
    z_₁ → z,
   {"Δ[y]" → Last[𝔼[0, 0, y₁] ~B₁~ bΔ₁→₁,₂],
    "Δ[b]" → Last[𝔼[0, 0, b₁] ~B₁~ bΔ₁→₁,₂],
    "Δ[a]" → Last[𝔼[0, 0, a₁] ~B₁~ aΔ₁→₁,₂],
    "Δ[x]" → Last[𝔼[0, 0, x₁] ~B₁~ aΔ₁→₁,₂]},
   {
     "S(a)" → ((𝔼[0, 0, a₁] ~B₁~ aS₁)⟦3⟧),
     "S(x)" → ((𝔼[0, 0, x₁] ~B₁~ aS₁)⟦3⟧),
     "S(b)" → ((𝔼[0, 0, b₁] ~B₁~ bS₁)⟦3⟧),
     "S(y)" → ((𝔼[0, 0, y₁] ~B₁~ bS₁)⟦3⟧)
    } /. z_₁ → z}
```

Out[∘]= {0.625,

$\{\{[a,x] \to -x,\ [b,y] \to -y\,\epsilon + O[\epsilon]^3\},\ \{\Delta[y] \to (B_2\, y_1 + y_2) + O[\epsilon]^3,\ \Delta[b] \to (b_1 + b_2) + O[\epsilon]^3,$

$\Delta[a] \to (a_1 + a_2) + O[\epsilon]^3,\ \Delta[x] \to (x_1 + x_2) - a_1\, x_2\, \epsilon + \frac{1}{2} a_1^2\, x_2\, \epsilon^2 + O[\epsilon]^3\},\ \{S(a) \to -a + O[\epsilon]^3,$

$S(x) \to -x - a\, x\, \epsilon - \frac{1}{2}(a^2\, x)\,\epsilon^2 + O[\epsilon]^3,\ S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + O[\epsilon]^3\}\}\}$

Hopf algebra axioms on both sides separately.

Associativity of am and bm:

```
In[∘]:= Timing@Block[{$k = 3},
   HL /@ {(am₁,₂→₁ ~B₁~ am₁,₃→₁) ≡ (am₂,₃→₂ ~B₂~ am₁,₂→₁), (bm₁,₂→₁ ~B₁~ bm₁,₃→₁) ≡ (bm₂,₃→₂ ~B₂~ bm₁,₂→₁)}
  ]
```

Out[∘]= {0.1875, {True, True}}

R and *P* are inverses:

```
In[∘]:= Timing@Block[{$k = 3}, {Rᵢ,ⱼ, Pᵢ,ₖ, HL[Rᵢ,ⱼ ~Bᵢ~ Pᵢ,ₖ ≡ 𝔼[aⱼ αₖ, xⱼ ξₖ, 1]]}]
```

Out[∘]= {0.109375, {𝔼[aⱼ bᵢ, xⱼ yᵢ,

$1 - \frac{1}{4}(x_j^2\, y_i^2)\,\epsilon + \left(\frac{1}{9} x_j^3\, y_i^3 + \frac{1}{32} x_j^4\, y_i^4\right)\epsilon^2 + \frac{(24\, x_j^2\, y_i^2 - 72\, x_j^4\, y_i^4 - 32\, x_j^5\, y_i^5 - 3\, x_j^6\, y_i^6)\,\epsilon^3}{1152} + O[\epsilon]^4],$

$\mathbb{E}[\alpha_k\, \beta_i,\ \eta_i\, \xi_k,\ 1 + \frac{1}{4}\eta_i^2\, \xi_k^2\,\epsilon + \frac{1}{288}(36\,\eta_i^2\,\xi_k^2 + 40\,\eta_i^3\,\xi_k^3 + 9\,\eta_i^4\,\xi_k^4)\,\epsilon^2 + \frac{1}{1152}$

$(48\,\eta_i^2\,\xi_k^2 + 192\,\eta_i^3\,\xi_k^3 + 156\,\eta_i^4\,\xi_k^4 + 40\,\eta_i^5\,\xi_k^5 + 3\,\eta_i^6\,\xi_k^6)\,\epsilon^3 + O[\epsilon]^4],$ True}}

as and $\overline{\text{aS}}$ are inverses, bs and $\overline{\text{bS}}$ are inverses:

```
In[∘]:= Timing[HL /@ {aS̅₁ ~B₁~ aS₁ ≡ 𝔼[a₁ α₁, x₁ ξ₁, 1], bS̅₁ ~B₁~ bS₁ ≡ 𝔼[b₁ β₁, y₁ η₁, 1]}]
```

Out[∘]= {0.390625, {True, True}}

(co)-associativity on both sides

In[∘]:= $\text{Timing}[\text{HL} /@ \{(a\Delta_{1\to1,2} \sim B_2 \sim a\Delta_{2\to2,3}) \equiv (a\Delta_{1\to1,3} \sim B_1 \sim a\Delta_{1\to1,2}),\ (b\Delta_{1\to1,2} \sim B_2 \sim b\Delta_{2\to2,3}) \equiv (b\Delta_{1\to1,3} \sim B_1 \sim b\Delta_{1\to1,2}),\ (am_{1,2\to1} \sim B_1 \sim am_{1,3\to1}) \equiv (am_{2,3\to2} \sim B_2 \sim am_{1,2\to1}),\ (bm_{1,2\to1} \sim B_1 \sim bm_{1,3\to1}) \equiv (bm_{2,3\to2} \sim B_2 \sim bm_{1,2\to1})\}]$

Out[∘]= {0.296875, {True, True, True, True}}

Δ is an algebra morphism

In[∘]:= $\text{Timing}[\text{HL} /@ \{am_{1,2\to1} \sim B_1 \sim a\Delta_{1\to1,2} \equiv (a\Delta_{1\to1,3}\ a\Delta_{2\to2,4}) \sim B_{1,2,3,4} \sim (am_{3,4\to2}\ am_{1,2\to1}),\ bm_{1,2\to1} \sim B_1 \sim b\Delta_{1\to1,2} \equiv (b\Delta_{1\to1,3}\ b\Delta_{2\to2,4}) \sim B_{1,2,3,4} \sim (bm_{3,4\to2}\ bm_{1,2\to1})\}]$

Out[∘]= {0.46875, {True, True}}

S is convolution inverse of id

$\text{Timing}[\text{HL}[\# \equiv \mathbb{E}[0, 0, 1]] \,\&\, /@ \{(a\Delta_{1\to1,2} \sim B_1 \sim aS_1) \sim B_{1,2} \sim am_{1,2\to1},\ (a\Delta_{1\to1,2} \sim B_2 \sim aS_2) \sim B_{1,2} \sim am_{1,2\to1},\ (b\Delta_{1\to1,2} \sim B_1 \sim bS_1) \sim B_{1,2} \sim bm_{1,2\to1},\ (b\Delta_{1\to1,2} \sim B_2 \sim bS_2) \sim B_{1,2} \sim bm_{1,2\to1}\}]$

Out[∘]= {0.4375, {True, True, True, True}}

But not with the opposite product:

In[∘]:= $\text{Timing}[\text{Short}[\# \equiv \mathbb{E}[0, 0, 1]] \,\&\, /@ \{(a\Delta_{1\to1,2} \sim B_1 \sim aS_1) \sim B_{1,2} \sim am_{2,1\to1},\ (a\Delta_{1\to1,2} \sim B_2 \sim aS_2) \sim B_{1,2} \sim am_{2,1\to1},\ (b\Delta_{1\to1,2} \sim B_1 \sim bS_1) \sim B_{1,2} \sim bm_{2,1\to1},\ (b\Delta_{1\to1,2} \sim B_2 \sim bS_2) \sim B_{1,2} \sim bm_{2,1\to1}\}]$

Out[∘]= $\{0.265625, \{-\epsilon\, x_1\, \mathcal{A}_1\, \xi_1 == 0,\ -\epsilon\, x_1\, \xi_1 == 0,\ -\epsilon\, y_1\, \eta_1 == 0,\ -\frac{\epsilon\, y_1\, \eta_1}{B_1} == 0\}\}$

S is an algebra anti-(co)morphism

In[∘]:= $\text{Timing}[\text{HL} /@ \{am_{1,2\to1} \sim B_1 \sim aS_1 \equiv (aS_1\ aS_2) \sim B_{1,2} \sim am_{2,1\to1},\ bm_{1,2\to1} \sim B_1 \sim bS_1 \equiv (bS_1\ bS_2) \sim B_{1,2} \sim bm_{2,1\to1},\ aS_1 \sim B_1 \sim a\Delta_{1\to1,2} \equiv a\Delta_{1\to2,1} \sim B_{1,2} \sim (aS_1\ aS_2),\ bS_1 \sim B_1 \sim b\Delta_{1\to1,2} \equiv b\Delta_{1\to2,1} \sim B_{1,2} \sim (bS_1\ bS_2)\}]$

Out[∘]= {0.640625, {True, True, True, True}}

Pairing axioms

In[∘]:= $\text{Timing}[\text{HL} /@ \{(bm_{1,2\to1}\ \mathbb{E}[\alpha_3\, a_3,\ \xi_3\, x_3,\ 1]) \sim B_{1,3} \sim P_{1,3} \equiv (\mathbb{E}[\beta_1\, b_1,\ \eta_1\, y_1,\ 1]\ \mathbb{E}[\beta_2\, b_2,\ \eta_2\, y_2,\ 1]\ a\Delta_{3\to4,5}) \sim B_{1,4} \sim P_{1,4} \sim B_{2,5} \sim P_{2,5},\ (b\Delta_{1\to1,2}\ \mathbb{E}[\alpha_3\, a_3,\ \xi_3\, x_3,\ 1]\ \mathbb{E}[\alpha_4\, a_4,\ \xi_4\, x_4,\ 1]) \sim B_{1,3} \sim P_{1,3} \sim B_{2,4} \sim P_{2,4} \equiv (\mathbb{E}[\beta_1\, b_1,\ \eta_1\, y_1,\ 1]\ am_{3,4\to3}) \sim B_{1,3} \sim P_{1,3}\}]$

Out[∘]= {0.296875, {True, True}}

In[∘]:= $\text{Timing}[\text{HL} /@ \{(bS_1\ \mathbb{E}[\alpha_2\, a_2,\ \xi_2\, x_2,\ 1]) \sim B_{1,2} \sim P_{1,2} \equiv (\mathbb{E}[\beta_1\, b_1,\ \eta_1\, y_1,\ 1]\ aS_2) \sim B_{1,2} \sim P_{1,2},\ (\overline{bS}_1\ \mathbb{E}[\alpha_2\, a_2,\ \xi_2\, x_2,\ 1]) \sim B_{1,2} \sim P_{1,2} \equiv (\mathbb{E}[\beta_1\, b_1,\ \eta_1\, y_1,\ 1]\ \overline{aS}_2) \sim B_{1,2} \sim P_{1,2}\}]$

Out[∘]= {0.171875, {True, True}}

Tests for the double.

Check the double formulas on the generators agree with SL2Portfolio.pdf:

In[ ]:=
```
Timing@{{
    "[a,y]" → ((𝔼[0, 0, y2 a1] ~B1,2~ dm1,2→1)⟦3⟧ - (𝔼[0, 0, y1 a2] ~B1,2~ dm1,2→1)⟦3⟧),
    "[b,x]" → ((𝔼[0, 0, x2 b1] ~B1,2~ dm1,2→1)⟦3⟧ - (𝔼[0, 0, x1 b2] ~B1,2~ dm1,2→1)⟦3⟧),
    "xy-qyx" → ((𝔼[0, 0, x1 y2] ~B1,2~ dm1,2→1)⟦3⟧ - (1 + ϵ) (𝔼[0, 0, y1 x2] ~B1,2~ dm1,2→1)⟦3⟧)
   } /. {z_1 → z} // Expand // Factor,
  {
   "Δ(a)" → ((𝔼[0, 0, a1] ~B1~ dΔ1→1,2)⟦3⟧),
   "Δ(x)" → ((𝔼[0, 0, x1] ~B1~ dΔ1→1,2)⟦3⟧),
   "Δ(b)" → ((𝔼[0, 0, b1] ~B1~ dΔ1→1,2)⟦3⟧),
   "Δ(y)" → ((𝔼[0, 0, y1] ~B1~ dΔ1→1,2)⟦3⟧)
  } // Simplify,
  {
    "S(a)" → ((𝔼[0, 0, a1] ~B1~ dS1)⟦3⟧),
    "S(x)" → ((𝔼[0, 0, x1] ~B1~ dS1)⟦3⟧),
    "S(b)" → ((𝔼[0, 0, b1] ~B1~ dS1)⟦3⟧),
    "S(y)" → ((𝔼[0, 0, y1] ~B1~ dS1)⟦3⟧)
   } /. {z_1 → z} // Simplify
 }
```

Out[ ]= $\{5.34375,$

$\{\{[a,y] \to -y + O[\epsilon]^3,\ [b,x] \to x\,\epsilon + O[\epsilon]^3,\ xy\text{-}qyx \to (1-B) + a\,B\,\epsilon + \frac{1}{2}\left(-a^2 B + x\,y\right)\epsilon^2 + O[\epsilon]^3\},$

$\{\Delta(a) \to (a_1 + a_2) + O[\epsilon]^3,\ \Delta(x) \to (x_1 + x_2) - a_1 x_2\,\epsilon + \frac{1}{2} a_1^2 x_2\,\epsilon^2 + O[\epsilon]^3,$

$\Delta(b) \to (b_1 + b_2) + O[\epsilon]^3,\ \Delta(y) \to (y_1 + B_1 y_2) + O[\epsilon]^3\},\ \{S(a) \to -a + O[\epsilon]^3,$

$S(x) \to -x - a\,x\,\epsilon - \frac{1}{2}\left(a^2 x\right)\epsilon^2 + O[\epsilon]^3,\ S(b) \to -b + O[\epsilon]^3,\ S(y) \to -\frac{y}{B} + \frac{y\,\epsilon}{B} - \frac{y\,\epsilon^2}{2\,B} + O[\epsilon]^3\}\}\}$

(co)-associativity

In[ ]:=
```
Timing[HL /@
  {(dΔ1→1,2 ~B2~ dΔ2→2,3) ≡ (dΔ1→1,3 ~B1~ dΔ1→1,2), (dm1,2→1 ~B1~ dm1,3→1) ≡ (dm2,3→2 ~B2~ dm1,2->1)}]
```

Out[ ]= {4.1875, {True, True}}

Δ is an algebra morphism

In[ ]:=
```
Timing@HL[dm1,2→1 ~B1~ dΔ1→1,2 ≡ (dΔ1→1,3 dΔ2→2,4) ~B1,2,3,4~ (dm3,4→2 dm1,2→1)]
```

Out[ ]= {6.73438, True}

S is convolution inverse of id

In[ ]:=
```
Timing[
 HL[# ≡ 𝔼[0, 0, 1]] & /@ {(dΔ1→1,2 ~B1~ dS1) ~B1,2~ dm1,2→1, (dΔ1→1,2 ~B2~ dS2) ~B1,2~ dm1,2→1}]
```

Out[ ]= {6.48438, {True, True}}

S is a (co)-algebra anti-morphism

```
In[∘]:= Timing[HL /@
          Expand /@ {dm_{1,2→1} ~ B_1 ~ dS_1 ≡ (dS_1 dS_2) ~ B_{1,2} ~ dm_{2,1→1}, dS_1 ~ B_1 ~ dΔ_{1→1,2} ≡ dΔ_{1→2,1} ~ B_{1,2} ~ (dS_1 dS_2)}]
Out[∘]= {11.6406, {True, True}}
```

Quasi-triangular axiom 1:

```
In[∘]:= Timing@HL[R_{1,2} ~ B_1 ~ dΔ_{1->1,3} ≡ (R_{1,4} R_{3,2}) ~ B_{2,4} ~ dm_{2,4→2}]
Out[∘]= {0.40625, True}
```

Quasi-triangular axiom 2:

```
In[∘]:= Timing@HL[((dΔ_{1->1,2} R_{3,4}) ~ B_{1,2,3,4} ~ (dm_{1,3→1} dm_{2,4→2})) ≡ ((dΔ_{1→2,1} R_{3,4}) ~ B_{1,2,3,4} ~ (dm_{3,1→1} dm_{4,2→2}))]
Out[∘]= {5.26563, True}
```

The Drinfel'd element inverse property, $(u_1 \overline{u}_2) \sim B_{1,2} \sim \mathrm{dm}_{1,2\to 1} \equiv \mathbb{E}[0, 0, 1]$:

```
In[∘]:= Timing@
         HL[((R_{1,2} ~ B_1 ~ dS_1 ~ B_{1,2} ~ dm_{2,1→i}) (R_{1,2} ~ B_2 ~ dS_2 ~ B_2 ~ dS_2 ~ B_{1,2} ~ dm_{2,1→j})) ~ B_{i,j} ~ dm_{i,j→i} ≡ 𝔼[0, 0, 1]]
Out[∘]= {2.5, True}
```

The ribbon element v satisfies $v^2 = S(u)\,u$. The spinner C=uv^{-1}. It is convenient to compute $z = S(u)\,u^{-1}$ which is something easy.

```
In[∘]:= Timing@Block[{$k = 3},
          (((R_{1,2} ~ B_1 ~ dS_1 ~ B_{1,2} ~ dm_{2,1→i}) ~ B_i ~ dS_i) (R_{1,2} ~ B_2 ~ dS_2 ~ B_2 ~ dS_2 ~ B_{1,2} ~ dm_{2,1→j})) ~ B_{i,j} ~ dm_{i,j→i}]
```

$$\text{Out}= \left\{30.6719,\ \mathbb{E}\left[0,\ 0,\ \frac{1}{B_i} + \frac{a_i \epsilon}{B_i} + \frac{a_i^2 \epsilon^2}{2 B_i} + \frac{a_i^3 \epsilon^3}{6 B_i} + O[\epsilon]^4\right]\right\}$$

```
In[∘]:= Timing@Block[{$k = 2}, HL /@ {(C_i C̄_j) ~ B_{i,j} ~ dm_{i,j→i} ≡ 𝔼[0, 0, 1], (C̄_i C̄_j) ~ B_{i,j} ~ dm_{i,j→i} ≡
            (((R_{1,2} ~ B_1 ~ dS_1 ~ B_{1,2} ~ dm_{2,1→i}) ~ B_i ~ dS_i) (R_{1,2} ~ B_2 ~ dS_2 ~ B_2 ~ dS_2 ~ B_{1,2} ~ dm_{2,1→j})) ~ B_{i,j} ~ dm_{i,j→i}}]
Out[∘]= {2.82813, {True, True}}
```

Reidemeister 2:

```
In[∘]:= Timing[HL[# ≡ 𝔼[0, 0, 1]] & /@
          {(R̄_{1,2} R_{3,4}) ~ B_{1,2,3,4} ~ (dm_{1,3→1} dm_{2,4→2}), (R_{1,2} R̄_{3,4}) ~ B_{1,2,3,4} ~ (dm_{1,3→1} dm_{2,4→2})}]
Out[∘]= {3.5625, {True, True}}
```

Cyclic Reidemeister 2:

```
In[∘]:= Timing@HL[(R_{1,4} R̄_{5,2} C̄_3) ~ B_{2,4} ~ dm_{2,4→2} ~ B_{1,3} ~ dm_{1,3→1} ~ B_{1,5} ~ dm_{1,5→1} ≡ C̄_1]
Out[∘]= {1.48438, True}
```

Reidemeister 3:

```
In[∘]:= Timing@HL[((R_{1,2} R_{4,3} R_{5,6}) ~ B_{1,4} ~ dm_{1,4→1} ~ B_{2,5} ~ dm_{2,5→2} ~ B_{3,6} ~ dm_{3,6→3}) ≡
          ((R_{1,6} R_{2,3} R_{4,5}) ~ B_{1,4} ~ dm_{1,4→1} ~ B_{2,5} ~ dm_{2,5→2} ~ B_{3,6} ~ dm_{3,6→3})]
Out[∘]= {3.09375, True}
```

Relations between the four kinks:

```
In[]:= Timing[HL /@ {Kink_i ≡ (R_{3,1} C_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→i},
        \overline{Kink}_j ≡ (\overline{R}_{3,1} \overline{C}_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→j}, (Kink_i \overline{Kink}_j) ~ B_{i,j} ~ dm_{i,j→1} ≡ 𝔼[0, 0, 1]}]
Out[]= {3.1875, {True, True, True}}
```

The Trefoil

```
In[]:= Timing@Block[{$k = 1},
        Z = R_{1,5} R_{6,2} R_{3,7} \overline{C}_4 \overline{Kink}_8 \overline{Kink}_9 \overline{Kink}_{10};
        Do[Z = Z ~ B_{1,r} ~ dm_{1,r→1}, {r, 2, 10}];
        {Simplify /@ Z, Simplify /@ (Z ~ B_1 ~ b2t_1 /. T_1 → T)}]
```

Out[]= $\{8.95313,$

$$\left\{\mathbb{E}\left[0,0,\frac{B_1}{1-B_1+B_1^2}+\left(B_1\left(-B_1+2B_1^2+2B_1^4+a_1\left(-1+B_1-B_1^3+B_1^4\right)-2x_1y_1-B_1^3\left(3+2x_1y_1\right)\right)\epsilon\right)\Big/\left(1-B_1+B_1^2\right)^3+O[\epsilon]^2\right],\ \mathbb{E}\left[0,0,\frac{T}{1-T+T^2}+\left(T\left(T\left(-1+2T-3T^2+2T^3\right)+2\left(-1+T-T^3+T^4\right)a_1-2\left(1+T^3\right)x_1y_1\right)\epsilon\right)\Big/\left(1-T+T^2\right)^3+O[\epsilon]^2\right]\right\}\}$$

Program

```
In[]:= Define[kR_{i,j} = R_{i,j} ~ B_{i,j} ~ (b2t_i b2t_j) /. t_{i|j} → t,
        \overline{kR}_{i,j} = \overline{R}_{i,j} ~ B_{i,j} ~ (b2t_i b2t_j) /. {t_{i|j} → t, T_{i|j} → T},
        km_{i,j→k} = (t2b_i t2b_j) ~ B_{i,j} ~ dm_{i,j→k} ~ B_k ~ b2t_k /. {t_k → t, T_k → T, τ_{i|j} → 0},
        kC_i = C_i ~ B_i ~ b2t_i /. T_i → T,
        \overline{kC}_i = \overline{C}_i ~ B_i ~ b2t_i /. T_i → T,
        kKink_i = Kink_i ~ B_i ~ b2t_i /. {t_i → t, T_i → T},
        \overline{kKink}_i = \overline{Kink}_i ~ B_i ~ b2t_i /. {t_i → t, T_i → T}]
```

```
In[]:= Timing@Block[{$k = 1},
        Z = kR_{1,5} kR_{6,2} kR_{3,7} \overline{kC}_4 \overline{kKink}_8 \overline{kKink}_9 \overline{kKink}_{10};
        Do[Z = Z ~ B_{1,r} ~ km_{1,r→1}, {r, 2, 10}];
        Simplify /@ Z]
```

Out[]=
$$\left\{4.4375,\ \mathbb{E}\left[0,0,\frac{T}{1-T+T^2}+\left(T\left(T\left(-1+2T-3T^2+2T^3\right)+2\left(-1+T-T^3+T^4\right)a_1-2\left(1+T^3\right)x_1y_1\right)\epsilon\right)\Big/\left(1-T+T^2\right)^3+O[\epsilon]^2\right]\right\}$$

RVK, rot, Z from 2016-09/OneSmidgen.nb. See also local version in this folder.

Program

```
In[]:= RVK::usage =
        "RVK[xs, rots] represents a Rotational Virtual Knot with a list of n Xp/Xm crossings
          xs and a length 2n list of rotation numbers rots. Crossing
          sites are indexed 1 through 2n, and rots⟦k⟧ is the rotation
          between site k-1 and site k. RVK is also a casting operator
          converting to the RVK presentation from other knot presentations.";
```

Program

In[ ]:=
```
RVK[pd_PD] := PP_RVK@Module[{n, xs, x, rots, front, k},
    n = Length[pd];
    xs = List @@ pd /. x_X :> If[PositiveQ[x], Xp[x[[4]], x[[1]]], Xm[x[[2]], x[[1]]]];
    rots = Table[0, {2 n}];
    front = {0};
    For[k = 0, k < 2 n, ++k,
     If[k == 0 ∨ FreeQ[front, -k],
      front = Flatten[front /. k → Catch[xs /. {
              Xp[k + 1, l_] | Xm[l_, k + 1] :> Throw[{l, k + 1, 1 - l}],
              Xp[l_, k + 1] | Xm[k + 1, l_] :> (++rots[[l]]; Throw[{1 - l, k + 1, l}])
             }]],
      If[MatchQ[front, {___, k, ___, -k, ___}], --rots[[k + 1]]]
     ]
    ];
    RVK[xs, rots]
   ];
RVK[K_] := RVK[PD[K]];
```

In[ ]:=
```
RVK[Knot[3, 1]]
```

KnotTheory: Loading precomputed data in PD4Knots`.

Out[ ]= RVK[{Xm[4, 1], Xm[6, 3], Xm[2, 5]}, {0, 0, 0, -1, 0, 0}]

Program

In[ ]:=
```
rot[_, 0] = 𝔼[0, 0, 1];
rot[i_, n_Integer] /; n > 0 :=
  rot[i, n] = Module[{j}, (rot[i, n - 1] kC_j) ~B_{i,j}~ km_{i,j→i}];
rot[i_, n_Integer] /; n < 0 := rot[i, n] = Module[{j}, (rot[i, n + 1] \overline{kC}_j) ~B_{i,j}~ km_{i,j→i}];
```

In[ ]:=
```
rot[i, -3]
```

Out[ ]= $\mathbb{E}\left[0, 0, \frac{1}{T^{3/2}} + \frac{3\, a_i\, \epsilon}{T^{3/2}} + \frac{9\, a_i^2\, \epsilon^2}{2\, T^{3/2}} + O[\epsilon]^3\right]$

Program

```
In[•]:= Z[K_] := Z[RVK@K];
Z[rvk_RVK] := (*Z[rvk] =*)
 PP"Z"@Module[{todo, n, rots, ζ, done, st, cx, ζ1, i, j, k, k1, k2, k3},
   {todo, rots} = List @@ rvk;
   AppendTo[rots, 0];
   n = Length[todo];
   ζ = 𝔼[0, 0, 1];
   done = {0};
   st = Range[0, 2 n + 1];
   While[todo =!= {},
    {cx} = MaximalBy[todo, Length[done ⋂ {#[[1]], #[[2]], #[[1]] - 1, #[[2]] - 1}] &, 1];
    {i, j} = List @@ cx;
    ζ1 = Switch[Head[cx],
       Xp, (kR_{i,j} OverBar[kKink_k]) ~ B_{j,k} ~ km_{j,k→j},
       Xm, (OverBar[kR_{i,j}] kKink_k) ~ B_{j,k} ~ km_{j,k→j}
      ];
    ζ1 = (rot[k, rots[[i]]] ζ1) ~ B_{k,i} ~ km_{k,i→i}; rots[[i]] = 0;
    ζ1 = (ζ1 rot[k, rots[[i + 1]]]) ~ B_{i,k} ~ km_{i,k→i}; rots[[i + 1]] = 0;
    ζ1 = (rot[k, rots[[j]]] ζ1) ~ B_{k,j} ~ km_{k,j→j}; rots[[j]] = 0;
    ζ1 = (ζ1 rot[k, rots[[j + 1]]]) ~ B_{j,k} ~ km_{j,k→j}; rots[[j + 1]] = 0;
    ζ *= ζ1;
    If[MemberQ[done, i], ζ = ζ ~ B_{i,i+1} ~ km_{i,i+1→i}; st = st /. st[[i + 2]] → st[[i + 1]]];
    If[MemberQ[done, i - 1], ζ = ζ ~ B_{st[[i]],i} ~ km_{st[[i]],i→st[[i]]}; st = st /. st[[i + 1]] → st[[i]]];
    If[MemberQ[done, j], ζ = ζ ~ B_{j,j+1} ~ km_{j,j+1→j}; st = st /. st[[j + 2]] → st[[j + 1]]];
    If[MemberQ[done, j - 1], ζ = ζ ~ B_{st[[j]],j} ~ km_{st[[j]],j→st[[j]]}; st = st /. st[[j + 1]] → st[[j]]];
    done = done ⋃ {i - 1, i, j - 1, j};
    todo = DeleteCases[todo, cx]
   ];
   Simplify /@ (ζ /. {x_0 → x, y_0 → y, a_0 → a})
  ]
```

Knot

```
In[•]:= $k = 1; Timing@Z@Knot[10, 100]
```

Knot

Out[•]= $\{902.922, \mathbb{E}[0, 0, T^4 / (1 - 4 T + 9 T^2 - 12 T^3 + 13 T^4 - 12 T^5 + 9 T^6 - 4 T^7 + T^8) + (T^4 (-6 + 2 T^{16} + 4 a (-2 + 14 T - 51 T^2 + 120 T^3 - 203 T^4 + 258 T^5 - 246 T^6 + 152 T^7 - 152 T^9 + 246 T^{10} - 258 T^{11} + 203 T^{12} - 120 T^{13} + 51 T^{14} - 14 T^{15} + 2 T^{16}) - 8 x y - 440 T^9 (-1 + x y) - 4 T^{15} (3 + 2 x y) + 8 T^8 (-97 + 21 x y) + 8 T^7 (131 + 21 x y) - 20 T^6 (57 + 22 x y) + T^{14} (37 + 48 x y) + T (44 + 48 x y) - 8 T^{11} (2 + 61 x y) + 8 T^5 (127 + 68 x y) - 2 T^{13} (35 + 78 x y) + 4 T^{10} (-39 + 136 x y) - T^2 (167 + 156 x y) + T^{12} (79 + 324 x y) + T^3 (410 + 324 x y) - T^4 (733 + 488 x y)) \epsilon) / (1 - 4 T + 9 T^2 - 12 T^3 + 13 T^4 - 12 T^5 + 9 T^6 - 4 T^7 + T^8)^3 + O[\epsilon]^2]\}$

```
In[•]:= EndProfile[];
```

Profile

```
In[•]:= PrintProfile[]
```

Profile

```
Out[•]= ProfileRoot is root. Profiled time: 1009.92
  (     1)     0.095/ 902.906 above Z
```

```
  (   149)    1.000/  82.530 above Bind
  (   126)    0.016/   0.032 above CF
  (     2)    0.016/   0.016 above RVK
  (    16)    0.047/   2.547 above Boot[1]
  (    18)    0.110/   6.421 above Boot[2]
  (     5)    0.077/  15.468 above Boot[3]
Exp: called 254335 times, time in 573.897/638.994
  ( 254335)  573.897/ 638.994 under Together
  ( 252433)   36.068/  65.097 above CF
QZip: called 296 times, time in 197.207/806.899
  (   296)  197.207/ 806.899 under Bind
  (   888)    0.799/ 575.501 above CF
  (   296)   11.190/  34.191 above Zip
Together: called 254335 times, time in 70.796/709.79
  ( 254335)   70.796/ 709.790 under CF
  ( 254335)  573.897/ 638.994 above Exp
Zip: called 2625 times, time in 56.547/194.735
  (   296)    7.069/  33.799 under LZip
  (   296)   11.190/  34.191 under QZip
  (  2033)   38.288/ 126.745 under Zip
  (  2625)   11.443/  11.443 above Collect
  (  2033)   38.288/ 126.745 above Zip
CF: called 254335 times, time in 53.955/763.745
  ( 252433)   36.068/  65.097 under Exp
  (   888)   17.072/ 123.115 under LZip
  (   126)    0.016/   0.032 under ProfileRoot
  (   888)    0.799/ 575.501 under QZip
  ( 254335)   70.796/ 709.790 above Together
LZip: called 296 times, time in 44.182/201.096
  (   296)   44.182/ 201.096 under Bind
  (   888)   17.072/ 123.115 above CF
  (   296)    7.069/  33.799 above Zip
Collect: called 2625 times, time in 11.443/11.443
  (  2625)   11.443/  11.443 under Zip
Bind: called 296 times, time in 1.422/1009.42
  (    72)    0.266/ 902.733 under Z
  (   149)    1.000/  82.530 under ProfileRoot
  (    32)    0.031/   2.547 under Boot[1]
  (    27)    0.062/   6.311 under Boot[2]
  (    16)    0.063/  15.296 under Boot[3]
  (   296)   44.182/ 201.096 above LZip
  (   296)  197.207/ 806.899 above QZip
Boot[3]: called 11 times, time in 0.172/26.313
  (     5)    0.077/  15.468 under ProfileRoot
  (     6)    0.095/  10.845 under Boot[3]
  (    16)    0.063/  15.296 above Bind
  (     6)    0.095/  10.845 above Boot[3]
Boot[2]: called 22 times, time in 0.11/6.437
  (    18)    0.110/   6.421 under ProfileRoot
  (     4)        0/   0.016 under Boot[2]
  (    27)    0.062/   6.311 above Bind
```

```
  (     4)       0/   0.016 above Boot[2]
Z: called 1 times, time in 0.095/902.906
  (     1)    0.095/ 902.906 under ProfileRoot
  (    72)    0.266/ 902.733 above Bind
  (     3)       0/   0.078 above Boot[1]
Boot[1]: called 27 times, time in 0.078/3.406
  (     3)       0/   0.078 under Z
  (    16)    0.047/   2.547 under ProfileRoot
  (     8)    0.031/   0.781 under Boot[1]
  (    32)    0.031/   2.547 above Bind
  (     2)       0/       0 above Boot[0]
  (     8)    0.031/   0.781 above Boot[1]
RVK: called 2 times, time in 0.016/0.016
  (     2)    0.016/   0.016 under ProfileRoot
Boot[0]: called 2 times, time in 0./0.
  (     2)       0/       0 under Boot[1]
```